<?xml version="1.0" encoding="UTF-8"?>
<!DOCTYPE xsl:stylesheet [
    <!ENTITY nbsp "&#xA0;">
    <!ENTITY nl "&#xD;&#xA;">
]>

<!-- stylesheet for mimetype "application/vnd.oasis.opendocument.presentation" -->

<!-- Version and namespaces -->
<xsl:stylesheet version="1.0"
    xmlns="http://www.w3.org/1999/xhtml"
    xmlns:anim="urn:oasis:names:tc:opendocument:xmlns:animation:1.0"
    xmlns:chart="urn:oasis:names:tc:opendocument:xmlns:chart:1.0"
    xmlns:config="urn:oasis:names:tc:opendocument:xmlns:config:1.0"
    xmlns:dc="http://purl.org/dc/elements/1.1/"
    xmlns:dr3d="urn:oasis:names:tc:opendocument:xmlns:dr3d:1.0"
    xmlns:draw="urn:oasis:names:tc:opendocument:xmlns:drawing:1.0"
    xmlns:fo="urn:oasis:names:tc:opendocument:xmlns:xsl-fo-compatible:1.0"
    xmlns:form="urn:oasis:names:tc:opendocument:xmlns:form:1.0"
    xmlns:int="http://opendocumentfellowship.org/internal"
    xmlns:math="http://www.w3.org/1998/Math/MathML"
    xmlns:meta="urn:oasis:names:tc:opendocument:xmlns:meta:1.0"
    xmlns:number="urn:oasis:names:tc:opendocument:xmlns:datastyle:1.0"
    xmlns:office="urn:oasis:names:tc:opendocument:xmlns:office:1.0"
    xmlns:ooo="http://openoffice.org/2004/office"
    xmlns:oooc="http://openoffice.org/2004/calc"
    xmlns:ooow="http://openoffice.org/2004/writer"
    xmlns:presentation="urn:oasis:names:tc:opendocument:xmlns:presentation:1.0"
    xmlns:script="urn:oasis:names:tc:opendocument:xmlns:script:1.0"
    xmlns:smil="urn:oasis:names:tc:opendocument:xmlns:smil-compatible:1.0"
    xmlns:style="urn:oasis:names:tc:opendocument:xmlns:style:1.0"
    xmlns:svg="urn:oasis:names:tc:opendocument:xmlns:svg-compatible:1.0"
    xmlns:table="urn:oasis:names:tc:opendocument:xmlns:table:1.0"
    xmlns:text="urn:oasis:names:tc:opendocument:xmlns:text:1.0"
    xmlns:xforms="http://www.w3.org/2002/xforms"
    xmlns:xlink="http://www.w3.org/1999/xlink"
    xmlns:xsl="http://www.w3.org/1999/XSL/Transform"
    exclude-result-prefixes="anim chart config dc dr3d draw fo form int math meta number
        office ooo oooc ooow presentation script smil style svg table text xforms xlink #default">

<!-- - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = -->

    <!-- import the base stylesheet -->
    <xsl:import href="common.xsl" />

    <!-- define the mimetype for this document -->
    <xsl:variable name="mimetype">application/vnd.oasis.opendocument.presentation</xsl:variable>

    <!-- represents the presentation's body -->
    <xsl:variable name="presentationBody" select="$documentContent/office:body/office:presentation" />

    <!-- represents the presentation's master page(s) -->
    <xsl:variable name="masterPage" select="$documentStyles/office:master-styles/style:master-page" />

<!-- - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = -->

<!-- PRESENTATION-SPECIFIC templates -->

    <xsl:template match="style:page-layout">
        <xsl:text>&nl;.</xsl:text>
        <xsl:call-template name="classInheritance">
            <xsl:with-param name="styleName"><xsl:value-of select="@style:name" /></xsl:with-param>
            <xsl:with-param name="class"><xsl:value-of select="translate(@style:name,'.,;: %()[]/\+','_____________')" /></xsl:with-param>
            <xsl:with-param name="separator"><xsl:text>.</xsl:text></xsl:with-param><!-- implement Combined Class Selector -->
        </xsl:call-template>
        <xsl:text> { position: relative; </xsl:text>
        <xsl:apply-templates select="style:page-layout-properties/@*" mode="styleattr" />
        <xsl:text>} </xsl:text>
    </xsl:template>

    <xsl:template name="odp-js">
        <xsl:element name="script">
            <xsl:attribute name="type"><xsl:text>application/javascript</xsl:text></xsl:attribute>
            <xsl:text>
    function showAll() {
        var pages = document.getElementsByName('pageBlock');
        for (i=0, j=1; i != pages.length; i++, j++) {
            document.getElementById('page'+j+'Header').style.display = 'none';
            document.getElementById('page'+j+'Links').style.display = '';
            document.getElementById('page'+j).style.display = '';
        }
        document.getElementById("linkAll").style.display = 'none';
    }
    function hideAll() {
        var pages = document.getElementsByName('pageBlock');
        for (i=0, j=1; i != pages.length; i++, j++) {
            pages[i].style.display = 'none';
        }
    }
    function showPage(pageNum) {
        hideAll();
        document.getElementById('page'+pageNum).style.display = '';
        document.getElementById('page'+pageNum+'Header').style.display = '';
        document.getElementById('page'+pageNum+'Links').style.display = 'none';
        document.getElementById("linkAll").style.display = '';
    }
    function gotoPage(displayType, pageNum) {
        if (displayType == "single") {
            var slideNumText = document.getElementById('singleGotoText'+pageNum);
            if (document.getElementById('page'+slideNumText.value) != null) {
                showPage(slideNumText.value);
                slideNumText.value = "";
            } else {
                alert("Please check the given slide #.");
            }
        } else if (displayType == "all") {
            var slideNumText = document.getElementById('allGotoText'+pageNum);
            if (document.getElementById('page'+slideNumText.value) != null) {
                window.location='#page'+slideNumText.value;
                slideNumText.value = "";
            } else {
                alert("Please check the given slide #.");
            }
        }
    }
            </xsl:text>
        </xsl:element>
    </xsl:template>

<!-- - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = -->

    <!-- Generate the actual pages (slides) - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - -->

    <xsl:template match="draw:page">
        <xsl:variable name="pageNum" select="presentation:notes/draw:page-thumbnail/@draw:page-number" />
        <xsl:variable name="pageLast" select="$presentationBody/draw:page[last()]/presentation:notes/draw:page-thumbnail/@draw:page-number" />
        <xsl:variable name="thisStyleName" select="./@draw:style-name" />
        <xsl:variable name="thisStyleProperties" select="$documentContent/office:automatic-styles/style:style[@style:name=$thisStyleName][1]/style:drawing-page-properties" />

        <div id="{@draw:name}" name="pageBlock">
            <!-- displays the headers/links for this slide -->
            <span id="{concat(@draw:name, 'Header')}" style="font-family:Arial;font-size:10pt;">
                <xsl:call-template name="createLinks">
                    <xsl:with-param name="prefix"><xsl:text>showPage(</xsl:text></xsl:with-param>
                    <xsl:with-param name="suffix"><xsl:text>)</xsl:text></xsl:with-param>
                    <xsl:with-param name="pageNum"><xsl:value-of select="$pageNum" /></xsl:with-param>
                    <xsl:with-param name="pageLast"><xsl:value-of select="$pageLast" /></xsl:with-param>
                    <xsl:with-param name="displayType"><xsl:text>single</xsl:text></xsl:with-param>
                </xsl:call-template>
            </span>
            <span id="{concat(@draw:name, 'Links')}" style="font-family:Arial;font-size:10pt;z-index:20;position:absolute;">
                <a onclick="showPage({$pageNum})" style="font-family:Arial;font-size:10pt;">Per Slide Display</a>
                <xsl:text> &nbsp; :: &nbsp; </xsl:text>
                <xsl:call-template name="createLinks">
                    <xsl:with-param name="prefix"><xsl:text>parent.location='#page</xsl:text></xsl:with-param>
                    <xsl:with-param name="suffix"><xsl:text>'</xsl:text></xsl:with-param>
                    <xsl:with-param name="pageNum"><xsl:value-of select="$pageNum" /></xsl:with-param>
                    <xsl:with-param name="pageLast"><xsl:value-of select="$pageLast" /></xsl:with-param>
                    <xsl:with-param name="displayType"><xsl:text>all</xsl:text></xsl:with-param>
                </xsl:call-template>
            </span><br />

            <xsl:variable name="masterStyle" select="@draw:master-page-name" />
            <xsl:variable name="pageLayout" select="$masterPage[@style:name=$masterStyle][1]/@style:page-layout-name" />
            <xsl:variable name="drawStyle" select="$masterPage[@style:name=$masterStyle][1]/@draw:style-name" />
            <!-- displays the actual slide -->
            <xsl:element name="div">
                <xsl:attribute name="class">
                    <xsl:call-template name="callClassInheritance">
                        <xsl:with-param name="paramValue"><xsl:value-of select="$masterStyle" /></xsl:with-param>
                    </xsl:call-template>
                    <xsl:text> </xsl:text>
                    <xsl:call-template name="callClassInheritance">
                        <xsl:with-param name="paramValue"><xsl:value-of select="$pageLayout" /></xsl:with-param>
                    </xsl:call-template>
                    <xsl:text> </xsl:text>
                    <xsl:call-template name="callClassInheritance">
                        <xsl:with-param name="paramValue"><xsl:value-of select="$drawStyle" /></xsl:with-param>
                    </xsl:call-template>
                    <xsl:text> </xsl:text>
                    <xsl:call-template name="callClassInheritance">
                        <xsl:with-param name="paramValue"><xsl:value-of select="@draw:style-name" /></xsl:with-param>
                    </xsl:call-template>
                </xsl:attribute>

                <!-- put the slide contents on top of the background image -->
                <xsl:attribute name="style">z-index: 5; background-image: none;</xsl:attribute>

                <!-- display the background image -->
                <xsl:choose>
                    <xsl:when test="$masterPage/draw:frame/draw:image">
                        <xsl:call-template name="displayBackground" />
                    </xsl:when>
                    <xsl:otherwise>
                        <xsl:element name="img">
                            <xsl:attribute name="src">
                                <xsl:value-of select="concat($param_imgloc, $documentStyles/office:styles/draw:fill-image[1]/@xlink:href)" />
                            </xsl:attribute>
                            <xsl:attribute name="style">width: 100%; height: 100%; z-index: -1; position: absolute;</xsl:attribute>
                        </xsl:element>
                    </xsl:otherwise>
                </xsl:choose>

                <!-- displays the slide contents -->
                <xsl:apply-templates />

                <!-- displays the header area -->
                <xsl:if test="$thisStyleProperties/@presentation:display-header='true'">
                    <xsl:call-template name="createPageAttrElement">
                        <xsl:with-param name="currentFrame" select="$masterPage/draw:frame[@presentation:class='header']" />
                        <xsl:with-param name="tempParam" select="@presentation:use-header-name" />
                    </xsl:call-template>
                </xsl:if>

                <!-- displays the footer area -->
                <xsl:if test="$thisStyleProperties/@presentation:display-footer='true'">
                    <xsl:call-template name="createPageAttrElement">
                        <xsl:with-param name="currentFrame" select="$masterPage/draw:frame[@presentation:class='footer']" />
                        <xsl:with-param name="tempParam" select="@presentation:use-footer-name" />
                    </xsl:call-template>
                </xsl:if>

                <!-- displays the date-time area -->
                <xsl:if test="$thisStyleProperties/@presentation:display-date-time='true'">
                    <xsl:call-template name="createPageAttrElement">
                        <xsl:with-param name="currentFrame" select="$masterPage/draw:frame[@presentation:class='date-time']" />
                        <xsl:with-param name="tempParam" select="@presentation:use-date-time-name" />
                    </xsl:call-template>
                </xsl:if>

                <!-- displays the slide number area -->
                <xsl:if test="$thisStyleProperties/@presentation:display-page-number='true'">
                    <xsl:call-template name="createPageAttrElement">
                        <xsl:with-param name="currentFrame" select="$masterPage/draw:frame[@presentation:class='page-number']" />
                        <xsl:with-param name="tempParam" select="$pageNum" />
                    </xsl:call-template>
                </xsl:if>

            </xsl:element><br />
        </div>
        <xsl:text>&nl;</xsl:text>
    </xsl:template>

    <!-- create the navigation links -->
    <xsl:template name="createLinks">
        <xsl:param name="prefix" />
        <xsl:param name="suffix" />
        <xsl:param name="pageNum" />
        <xsl:param name="pageLast" />
        <xsl:param name="displayType" />
        <xsl:element name="a">
            <xsl:if test="$pageNum > 1">
                <xsl:attribute name="onclick">
                    <xsl:value-of select="$prefix"/>1<xsl:value-of select="$suffix"/>
                </xsl:attribute>
            </xsl:if>
            <xsl:text>Home</xsl:text>
        </xsl:element>
        <xsl:text> | </xsl:text>
        <xsl:element name="a">
            <xsl:if test="$pageNum > 1">
                <xsl:attribute name="onclick">
                    <xsl:value-of select="$prefix"/><xsl:value-of select="$pageNum - 1" /><xsl:value-of select="$suffix"/>
                </xsl:attribute>
            </xsl:if>
            <xsl:text>Previous</xsl:text>
        </xsl:element>
        <xsl:text> &nbsp;::&nbsp; </xsl:text>
        <xsl:element name="a">
            <xsl:if test="$pageLast > $pageNum">
                <xsl:attribute name="onclick">
                    <xsl:value-of select="$prefix"/><xsl:value-of select="$pageNum + 1" /><xsl:value-of select="$suffix"/>
                </xsl:attribute>
            </xsl:if>
            <xsl:text>Next</xsl:text>
        </xsl:element>
        <xsl:text> | </xsl:text>
        <xsl:element name="a">
            <xsl:if test="$pageLast > $pageNum">
                <xsl:attribute name="onclick">
                    <xsl:value-of select="$prefix"/><xsl:value-of select="$pageLast" /><xsl:value-of select="$suffix"/>
                </xsl:attribute>
            </xsl:if>
            <xsl:text>Last</xsl:text>
        </xsl:element>
        <xsl:text> &nbsp; [ </xsl:text><xsl:value-of select="$pageNum" />
        <xsl:text> / </xsl:text><xsl:value-of select="$pageLast" /><xsl:text> ] &nbsp; :: &nbsp; Go to Slide#</xsl:text>

        <xsl:element name="input">
            <xsl:attribute name="type"><xsl:text>text</xsl:text></xsl:attribute>
            <xsl:attribute name="size"><xsl:text>3</xsl:text></xsl:attribute>
            <xsl:attribute name="name"><xsl:value-of select="$displayType"/><xsl:text>GotoText</xsl:text><xsl:value-of select="$pageNum" /></xsl:attribute>
            <xsl:attribute name="id"><xsl:value-of select="$displayType"/><xsl:text>GotoText</xsl:text><xsl:value-of select="$pageNum" /></xsl:attribute>
            <xsl:attribute name="value"></xsl:attribute>
        </xsl:element>
        <xsl:element name="input">
            <xsl:attribute name="type"><xsl:text>button</xsl:text></xsl:attribute>
            <xsl:attribute name="name"><xsl:value-of select="$displayType"/><xsl:text>GotoButton</xsl:text><xsl:value-of select="$pageNum" /></xsl:attribute>
            <xsl:attribute name="id"><xsl:value-of select="$displayType"/><xsl:text>GotoButton</xsl:text><xsl:value-of select="$pageNum" /></xsl:attribute>
            <xsl:attribute name="value"><xsl:text>Submit</xsl:text></xsl:attribute>
            <xsl:attribute name="onclick">
                <xsl:text>gotoPage('</xsl:text><xsl:value-of select="$displayType"/><xsl:text>','</xsl:text>
                <xsl:value-of select="$pageNum"/><xsl:text>')</xsl:text>
            </xsl:attribute>
        </xsl:element>
    </xsl:template>

    <!-- create the element to display the slide's header/footer/date-time/page-number -->
    <xsl:template name="createPageAttrElement">
        <xsl:param name="currentFrame" />
        <xsl:param name="tempParam" />
        <xsl:element name="div">
            <xsl:attribute name="class">
                <xsl:call-template name="callClassInheritance">
                    <xsl:with-param name="paramValue"><xsl:value-of select="$currentFrame/@presentation:style-name" /></xsl:with-param>
                </xsl:call-template>
                <xsl:text> </xsl:text>
                <xsl:call-template name="callClassInheritance">
                    <xsl:with-param name="paramValue"><xsl:value-of select="$currentFrame/@draw:text-style-name" /></xsl:with-param>
                </xsl:call-template>
            </xsl:attribute>
            <xsl:attribute name="style"><xsl:apply-templates select="$currentFrame" mode="attribs" /></xsl:attribute>
            <xsl:apply-templates select="$currentFrame/draw:text-box">
                <xsl:with-param name="tempParam" select="$tempParam" />
            </xsl:apply-templates>
        </xsl:element>
    </xsl:template>

    <!-- ignore the contents of the declaration items as they will be used elsewhere -->
    <xsl:template match="presentation:header-decl|presentation:footer-decl|presentation:date-time-decl|presentation:notes">
        <!-- do nothing for these elements -->
    </xsl:template>

    <!-- displays the header area content -->
    <xsl:template match="presentation:header">
        <xsl:param name="tempParam" />
        <xsl:value-of select="$presentationBody/presentation:header-decl[@presentation:name=$tempParam][1]" />
    </xsl:template>

    <!-- displays the footer area content -->
    <xsl:template match="presentation:footer">
        <xsl:param name="tempParam" />
        <xsl:value-of select="$presentationBody/presentation:footer-decl[@presentation:name=$tempParam][1]" />
    </xsl:template>

    <!-- displays the date-time area content -->
    <xsl:template match="presentation:date-time">
        <xsl:param name="tempParam" />
        <xsl:value-of select="$presentationBody/presentation:date-time-decl[@presentation:name=$tempParam][1]" />
    </xsl:template>

    <!-- displays the slide number area content -->
    <xsl:template match="text:page-number">
        <xsl:param name="tempParam" />
        <xsl:value-of select="translate(.,'&lt;number&gt;',$tempParam)" />
    </xsl:template>

    <!-- displays the background image (for customized backgrounds) -->
    <xsl:template name="displayBackground">
        <xsl:variable name="pageBckgrndFrame" select="$masterPage/draw:frame/draw:image" />
        <xsl:if test="$pageBckgrndFrame">
            <xsl:element name="div">
                <xsl:attribute name="class">
                    <xsl:call-template name="callClassInheritance">
                        <xsl:with-param name="paramValue"><xsl:value-of select="$pageBckgrndFrame/../@draw:style-name" /></xsl:with-param>
                    </xsl:call-template>
                    <xsl:text> </xsl:text>
                    <xsl:call-template name="callClassInheritance">
                        <xsl:with-param name="paramValue"><xsl:value-of select="$pageBckgrndFrame/../@draw:text-style-name" /></xsl:with-param>
                    </xsl:call-template>
                </xsl:attribute>
                <xsl:attribute name="style">
                    <xsl:if test="$pageBckgrndFrame/../@svg:width">width: <xsl:value-of select="$pageBckgrndFrame/../@svg:width" />; </xsl:if>
                    <xsl:if test="$pageBckgrndFrame/../@svg:height">height: <xsl:value-of select="$pageBckgrndFrame/../@svg:height" />; </xsl:if>
                    z-index: -2; position: absolute;
                </xsl:attribute>
                <xsl:element name="img">
                    <xsl:attribute name="src">
                        <xsl:value-of select="concat($param_imgloc, $pageBckgrndFrame/@xlink:href)" />
                    </xsl:attribute>
                    <xsl:attribute name="style">
                        width: 100%; height: 100%; z-index: -1; position: absolute;
                        <xsl:if test="$pageBckgrndFrame/../@svg:x">left: <xsl:value-of select="$pageBckgrndFrame/../@svg:x" />; </xsl:if>
                        <xsl:if test="$pageBckgrndFrame/../@svg:y">top: <xsl:value-of select="$pageBckgrndFrame/../@svg:y" />; </xsl:if>
                    </xsl:attribute>
                </xsl:element>
            </xsl:element>
        </xsl:if>
    </xsl:template>

<!-- - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = -->

<!-- GRAPHIC CONTENT - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - -->

    <!-- drawing the frame (or rectangle) -->
    <xsl:template match="draw:frame | draw:rect">
        <xsl:element name="div">
            <xsl:attribute name="class">
                <!-- only one between @draw:style-name and @presentation:style-name will exist at a time -->
                <xsl:if test="@draw:style-name">
                    <xsl:call-template name="callClassInheritance">
                        <xsl:with-param name="paramValue"><xsl:value-of select="@draw:style-name" /></xsl:with-param>
                    </xsl:call-template>
                </xsl:if>
                <xsl:if test="@presentation:style-name">
                    <xsl:call-template name="callClassInheritance">
                        <xsl:with-param name="paramValue"><xsl:value-of select="@presentation:style-name" /></xsl:with-param>
                    </xsl:call-template>
                </xsl:if>
                <xsl:if test="$mimetype=$odp-mime"> <!-- these item only applicable to presentation documents -->
                    <xsl:if test="@presentation:class">
                        <xsl:text> </xsl:text>
                        <xsl:call-template name="callClassInheritance">
                            <xsl:with-param name="paramValue"><xsl:value-of select="@presentation:class" /></xsl:with-param>
                        </xsl:call-template>
                    </xsl:if>
                </xsl:if>
                <xsl:if test="@draw:text-style-name">
                    <xsl:text> </xsl:text><xsl:value-of select="@draw:text-style-name" />
                </xsl:if>
            </xsl:attribute>
            <xsl:attribute name="style">
                <xsl:apply-templates select="." mode="attribs" />
            </xsl:attribute>
        <xsl:apply-templates />
        </xsl:element>
    </xsl:template>

    <!-- add the drawing element's properties in the inline CSS for this object -->
    <xsl:template match="draw:frame | draw:rect" mode="attribs">
        <xsl:if test="@svg:width">width: <xsl:value-of select="@svg:width" />; </xsl:if>
        <xsl:if test="@svg:height">height: <xsl:value-of select="@svg:height" />; </xsl:if>
        <xsl:if test="@svg:x">position: absolute; left: <xsl:value-of select="@svg:x" />; </xsl:if>
        <xsl:if test="@svg:y">top: <xsl:value-of select="@svg:y" />; </xsl:if>
        <xsl:if test="@fo:min-width">min-width: <xsl:value-of select="@fo:min-width" />; </xsl:if>
        <xsl:if test="@fo:min-height">min-height: <xsl:value-of select="@fo:min-height" />; </xsl:if>
        <xsl:if test="@fo:max-width">max-width: <xsl:value-of select="@fo:max-width" />; </xsl:if>
        <xsl:if test="@fo:max-height">max-height: <xsl:value-of select="@fo:max-height" />; </xsl:if>
        <xsl:if test="@draw:z-index">z-index: <xsl:value-of select="@draw:z-index" />; </xsl:if>
        <xsl:if test="not(@draw:z-index)">z-index: 10; </xsl:if>
        <xsl:if test="contains(draw:image/@xlink:href, 'ObjectReplacements')">background-color: transparent; </xsl:if>
    </xsl:template>

    <!-- IMAGES - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - -->

    <!-- template for displaying images -->
    <xsl:template match="draw:image">
        <xsl:element name="img">
            <xsl:if test="../svg:desc"><xsl:attribute name="alt">
                <xsl:value-of select="../svg:desc" />
            </xsl:attribute></xsl:if>
            <xsl:attribute name="src">
                <xsl:value-of select="concat($param_imgloc, @xlink:href)" />
            </xsl:attribute>
            <xsl:attribute name="style">
                <xsl:if test="../@svg:width">width: <xsl:value-of select="../@svg:width" />; </xsl:if>
                <xsl:if test="../@svg:height">height: <xsl:value-of select="../@svg:height" />; </xsl:if>
                <xsl:if test="../@draw:z-index">z-index: <xsl:value-of select="../@draw:z-index" />; </xsl:if>
                <xsl:if test="not(./@draw:z-index)">z-index: 0; </xsl:if>
            </xsl:attribute>
        </xsl:element>
    </xsl:template>

    <xsl:template match="svg:desc" />

    <!-- TEXT BOXES - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - -->

    <xsl:template match="draw:text-box">
        <xsl:param name="tempParam" />
        <xsl:element name="div">
            <xsl:apply-templates>
                <xsl:with-param name="tempParam" select="$tempParam" />
            </xsl:apply-templates>
        </xsl:element>
    </xsl:template>

<!-- - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = - = -->

<!-- others -->

    <xsl:template match="office:annotation|dc:creator|dc:date|meta:date-string">
        <!-- do not handle these tags for this mimetype, since by default, these annotation items are also not displayed to user -->
    </xsl:template>

</xsl:stylesheet>
